








Θέμα 1


Το πρόγραμμα αυτό είναι ένα χρονόμετρο που τρέχει στο προσκήνιο (foreground). Γίνεται χρήση των ρουτινών 03A0h kai 03ACh (DISPLAY) και της ρουτίνας 060Dh (DELAY). Το πρόγραμμα φορτώνεται στη διεύθυνση 2000h:








  MVI B,3CH


        MVI C,0AH


LABEL:  MVI A,00H


START:  STA 40F5H


        MVI A,0


LOOP1:  STA 40F4H


        MVI A,0


LOOP:   STA 40F6H


        PUSH PSW


        PUSH B


        CALL 03A0H


        CALL 03ACH


        LXI D,314DH


        CALL 060DH


        POP B


        POP PSW


        INR A


        DAA


        DCR C


        JNZ LOOP


        MVI C,0AH


        LDA 40F4H


        INR A


        DAA


        DCR B


        JNZ LOOP1


        MVI B,3CH


        LDA 40F5H


        INR A


        DAA


        JMP START


        RST 1


        END


�



Θέμα 2





Θα υλοποιήσουμε ένα χρονόμετρο λεπτών, δευτερολέπτων και δεκάτων στο 7-segment display που θα πρέπει να λειτουργεί ταυτόχρονα με το CRT monitor πρόγραμμα. Θα εφαρμόσουμε τη μέθοδο καταμερισμού του χρόνου χρησιμοποιώντας τονtimer 8155 του MIKROKIT


Συκεκριμένα, η χρήση του timer 8155 βασίζεται στην κλήση του RST 7.5. Αρχικοποιούμε τον timer στην τιμή 15720 παλμών για μηδενισμό και τον θέτουμε σε λειτουργία τύπου 10, όπου στέλνεται ένας μόνο παλμός RST 7.5 κατά το μηδενισμό του μετρητή. Εμείς γράφουμε την εξυπηρέτηση του RST 7.5 έτσι ώστε πριν απ'όλα ξαναρχικοποιεί και θέτει σε λειτουργία τον timer, ενώ κατόπιν αυξάνει τους εσωτερικούς μετρητές του προγράμματός μας. 


Για την εγγραφή των αριθμών χρησιμοποιούμε τις ρουτίνες 03A0h και 03ACh.


Το πρόγραμμα ξεκινά από τη διεύθυνση 2000h, όπου υπάρχει μια εντολή jump στο κυρίως σώμα της εξυπηρέτησης του RST 7.5 (το RST 7.5 στο ΜΙΚΡΟΚΙΤ κάνει jump στη διεύθυνση 2000h). Από τη διεύθυνση 2003h έχουμε τη ρουτίνα αρχικοποίησης (δίνουμε G2003 στο ΜΙΚΡΟΚΙΤ για να αρχίσει τη λειτουργία του το χρονόμετρο). Υπάρχει μια υπορουτίνα (TIMER) η οποία αναλαμβάνει την αρχικοποίηση και τη θέση σε λειτουργία του timer 8155 κάθε φορά που καλείται. Η ρουτίνα αυτή, όπως είπαμε, καλείται κάθε φορά που εξυπηρετείται το RST 7.5 και προετοιμάζει την επόμενη εξυπηρέτηση (μετά το πέρασμα των 0.005 sec).


Παρατίθεται η λίστα του προγράμματος:





;


;ask 5.2


;


;


        JMP R75 		; Jump to Main RST7.5 hander


START:  MVI A,0		; initialize


        STA BUF0		; zero counters


        STA HSEC


        STA SEC


        STA MIN


        STA 40F6H	; display on screen


        STA 40F5H


        STA 40F4H


        CALL 03A0H


        CALL 03ACH


        CALL TIMER	; initialize 8155





        RST 0;		; return to monitor


BUF0    DB 0		; count of 0.005 secs


HSEC    DB 0		; 1/10 sec.


SEC     DB 0		; sec.


MIN     DB 0		; min.





TIMER:  LDA 40FFH	; initialize timer


        ANI 3FH		; Get status and 


        ORI 40H		; stop timer


        OUT 40H	


        LXI B,15720D	; prepare for new timing


        MOV A,B		; for 0.005 sec


        ORI 80H


        OUT 45H		; out to the ports


        MOV A,C


        OUT 44H


        LDA 40FFH


        ORI C0H


        OUT 40H


        RET			; timer initialization ended





R75:  PUSH PSW		; handle RST7.5


        PUSH B		; save registers to stack


        PUSH D


        PUSH H


        CALL TIMER	; prepare timer for next time


        LDA BUF0


        INR A


        CPI 14H


        JZ L1


        STA BUF0


        POP H


        POP D


        POP B


        POP PSW


        RET





L1:     MVI A,0		; 1/10 of a second passed


        STA BUF0		; display


        LDA HSEC


        INR A


        CPI 0AH


        JZ L2


        STA HSEC


        STA 40F6H


        CALL 03ACH


        POP H


        POP D


        POP B


        POP PSW


        EI


        RET





L2:     MVI A,0		; 1 second passed


        STA HSEC		; display


        STA 40F6H


        CALL 03ACH


        LDA SEC


        INR A


        DAA


        CPI 60H


        JZ L3


        STA SEC


        STA 40F4H


        CALL 03A0H


        POP H


        POP D


        POP B


        POP PSW


        RET





L3:     MVI A,0		; 1 minute passed


        STA SEC		; display


        STA 40F4H


        CALL 03A0H


        LDA MIN


        INR A


        DAA


        STA MIN


        STA 40F5H


        CALL 03A0H


        POP H


        POP D


        POP B


        POP PSW


        RET			; end of RST7.5


        END





Tο παραπάνω πρόγραμμα εκτελείται δείχνοντας στην οθόνη το χρονόμετρο, ενώ στο τερματικό, από το PC βρισκόμαστε στο monitor του ΜΙΚΡΟΚΙΤ και μπορούμε να εκτελέσουμε οποιαδήποτε λειτουργία. Για δοκιμή, κατασκευάζουμε το παρακάτω πρόγραμμα, το οποίο είναι ένας μετρητής δευτερολέπτων, από 0 εως 9, ο οποίος χρησιμοποιεί τη ρουτίνα 060Dh για τη χρονοκαθυστέρηση. Η έξοδος του προγράμματος αυτού γίνεται στο τερματικό, ώστε να μην μπερδεύεται η έξοδός του με αυτή του προηγούμενου χρονομέτρου. Χρησιμοποιείται γι'αυτό το σκοπό η ρουτίνα 05FCh (έξοδος χαρακτήρα στη σειριακή). Τα δύο προγράμματα τρέχουν μαζί.


Το πρόγραμμα του μετρητη, που φορτώνεται στη θέση 2100h, είναι το:


 


        MVI A,0        


LOOP:   PUSH PSW


        ADI 30H


        MOV C,A


        CALL 05FCH


        MVI C,0DH


        CALL 05FCH


        MVI A,0AH





LOOP2:  LXI D,314DH


        PUSH PSW


        CALL 060DH


        POP PSW


        DCR A


        JNZ LOOP2


        POP PSW


        INR A


        CPI 0AH


        JNZ LOOP


        MVI A,0


        JMP LOOP


        END        

